



**Sterownik wyświetlacza
7-segmentowego
z licznikiem impulsów**
na płycie testowej z układem Spartan-3AN

Laboratorium Języków Opisu Sprzętu AGH WFIS

Wyświetlacz 7-segmentowy





Kod licznika modulo 10 z sygnalizacją przepełnienia:

```
timescale 1ns / 1ps
//////////////////////////////////////////////////////////////////
module cnt1dek(input clk, rst, en,
               output reg ovl, output reg[3:0] out);

always @(posedge clk, posedge rst)
    if(rst)
        out<=4'b0;
    else if(en)
        if(out==4'd9)
            out<=4'b0;
        else
            out<=out+1'b1;

always @(posedge clk, posedge rst)
    if(rst)
        ovl<=1'b0;
    else
        if(out==4'd9)
            ovl<=1'b1;
        else
            ovl<=1'b0;

endmodule
```

Kod licznika 3 dekadowego BCD z wykorzystaniem licznika modulo 10 z poprzedniego slajdu:

```
`timescale 1ns / 1ns

module cntBCD(input clk, rst, en, output ovl,
              output [11:0] out);

wire [3:0] out1;
wire [3:0] out2;
wire [3:0] out3;

cnt1dek bcd1 (clk, rst, en, ovl1, out1);
cnt1dek bcd2 (clk, rst, en & ovl1, ovl2, out2);
cnt1dek bcd3 (clk, rst, en & ovl1 & ovl2, ovl3, out3);

assign out = {out3,out2,out1};
assign ovl = ovl3;

endmodule
```

Kod o tej samej funkcjonalności co na poprzednim slajdzie ale o skalowalnej liczbie dekad:

```
`timescale 1ns / 1ns

module cntBCD(input clk, rst, en, output ovl,
              output [11:0] out);

    localparam s = 3;
    wire [s-1:0] ovll;
    wire [s:0] enn;
    genvar i;

    assign enn[0] = en;
    assign ovl = ovll[s-1];

    generate for (i=0; i<s; i = i + 1)
        begin: dek
            assign enn[i+1] = ovll[i] & enn[i];
            cnt1dek bcd1 (clk, rst, enn[i], ovll[i], out[4*i+3:4*i]);
        end
    endgenerate

endmodule
```

Kod detektora zbocza narastającego impulsu wejściowego:


```
module genPulse1(input clk, rst, trig,
                 output pulse);

reg [1:0] tmp;

always @(posedge clk, posedge rst)
    if(rst)
        tmp <= 2'b0;
    else begin
        tmp[0] <= trig;
        tmp[1] <= tmp[0];
    end

assign pulse = tmp[0] & ~tmp[1];

endmodule
```



Impulsy znacznie dłuższe od okresu zegara są zliczane tylko raz, krótsze od okresu zegara są ignorowane.

Kod dzielnika częstotliwości dostarczającego impulsu zezwolenia dla generatora adresu cyfry (anody):

```
module clkdiv #(parameter div=13'd5000) (input clk, rst, output reg clk_out);
    localparam N=clogb2(div);
    reg [N-1:0] counter;

    always @(posedge clk, posedge rst) begin
        if(rst)
            counter <= div;
        else
            if(counter == {N{1'b0}})
                counter <= div;
            else
                counter <= counter - 1'b1;
        end

    always @(posedge clk, posedge rst) begin
        if(rst)
            clk_out <= 1'b0;
        else
            if(counter == {N{1'b0}})
                clk_out <= 1'b1;
            else
                clk_out <= 1'b0;
        end

function integer clogb2;
input [31:0] value;
begin
    value = value - 1;
    for (clogb2 = 0; value > 0; clogb2 = clogb2 + 1)
        value = value >> 1;
end
endfunction
endmodule
```


Wskaźnik 7-segmentowy – złącze J2 na płycie Spartan-3AN

2	4	6	8	10	12	14	16	18	20	22	24	26	28	30	32	34
GN D	GN D	A0 (B4)	GN D	A2 (A5)	GN D	. (A6)	3.3V	3.3V	3.3V	f (A8)	GN D	d (C10)	GN D	b (A11)	GN D	GN D
GN D	GN D	A1 (A4)	GN D	A3 (B6)	GN D	g (A7)	3.3V	3.3V	3.3V	e (A9)	GN D	c (A10)	GN D	a (A12)	GN D	GN D
1	3	5	7	9	11	13	15	17	19	21	23	25	27	29	31	33

Plik ograniczeń projektowych UCF (User Constraint File) zapewniający poprawne podłączenie sygnałów do płytki wyświetlacza poprzez złącze J2:

```
#7seg; 7.11.2014 AS
CONFIG VCCAUX = "3.3" ;
CONFIG ENABLE_SUSPEND = "NO" ;
CONFIG POST_CRC = "DISABLE" ;

NET "clk"          LOC = "E12"   | IOSTANDARD = LVCMOS33 | PERIOD = 20.000 ;
OFFSET = IN  10.000 VALID 20.000 BEFORE "clk" ;
OFFSET = OUT 20.000 AFTER "clk" ;

NET "LEDs<0>" LOC = "A12"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #a
NET "LEDs<1>" LOC = "A11"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #b
NET "LEDs<2>" LOC = "A10"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #c
NET "LEDs<3>" LOC = "C10"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #d
NET "LEDs<4>" LOC = "A9"    | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #e
NET "LEDs<5>" LOC = "A8"    | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #f
NET "LEDs<6>" LOC = "A7"    | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #g

NET "anodes<0>" LOC = "B4"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #jdnosci
NET "anodes<1>" LOC = "A4"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #dziesiatki
NET "anodes<2>" LOC = "A5"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #setki
NET "anodes<3>" LOC = "B6"   | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ; #tysiace

NET "ov1"        LOC = „R20" | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ;
NET "en"         LOC = "V14" | IOSTANDARD = LVCMOS33 | DRIVE = 8 | SLEW = SLOW ;#generator
NET "rst"        LOC = "T14" | IOSTANDARD = LVCMOS33 | PULLDOWN ;#przycisk polnocny - reset
NET "sw"         LOC = "T16" | IOSTANDARD = LVCMOS33 | PULLDOWN ;#przycisk wschodni - zlicznie
```

Obserwacja działania układu

Impuls prostokątny:
Niski poziom 0V
Wysoki poziom 3V
Częstotliwość 5Hz

